

Homework 4: Declarative Concurrency

See Webcourses and the syllabus for due dates.

In this homework you will learn about the declarative concurrent model and basic techniques of programming in this model. The programming techniques include stream programming and lazy functional programming [Concepts] [UseModels]. A few problems also make comparisons with the declarative model and with concurrency features in Java [MapToLanguages].

A few problems also make comparisons between programming techniques [EvaluateModels]; you should look at Problem 26 on page 14 right now so you can be thinking about it while you do the other problems.

Answers to English questions should be in your own words; don't just quote text from the textbook.

Your code should be written in the declarative concurrent model, so you must not use cells and assignment in your Oz solutions. (Furthermore, note that the declarative model does *not* include the primitive `IsDet` or the library function `IsFree`; thus you are also prohibited from using either of these functions in your solutions, except where we explicitly allow them.) But please use all linguistic abstractions and syntactic sugars that are helpful.

For all Oz programming exercises, you must run your code using the Mozart/Oz system. For programming problems for which we provide tests, you can find them all in a zip file, which you can download from problem 1's assignment on Webcourses. If the tests don't pass, please try to say why they don't pass, as this enhances communication and makes commenting on the code easier and more specific to your problem.

Turn in (on Webcourses) your code and also turn in the output of your testing for all exercises that require code.

Please upload code as text files with the name given in the problem or testing file and with the suffix `.oz`. Please use the name of the main function as the name of the file. Please either paste into the webcourses answer box or upload test output and English answers; If you choose to upload test output or English answers as separate files, use plain text files with suffix `.txt` (or PDF files with suffix `.pdf`). If you have a mix of code and English, use a text file with a `.oz` file suffix, and put comments in the file for the English parts. (In any case, don't put any spaces in your file names!) Your code should compile with Oz, if it doesn't you probably should keep working on it. If you don't have time, at least tell us that you didn't get it to compile.

You should use helping functions whenever you find that useful. Unless we specifically say how you are to solve a problem, feel free to use any functions that are compatible with the declarative model from the Oz library (base environment), especially functions like `Map` and `Fo1dR`.

Your code should compile with Oz, if it doesn't you probably should keep working on it. If you don't have time, at least tell us that you didn't get it to compile.

You might lose points even if your code is passing all the tests, if you don't follow the grammar or if your code is confusing or needlessly inefficient. Make sure you don't have extra case clauses or unnecessary if tests, make sure you are using as many function definitions/calls as needed, and check if you are making unnecessary loops or passes in your code; elegant and efficient code earns full marks, working code does not necessarily earn full marks.

Don't hesitate to contact the staff if you are stuck at some point.

Read Chapter 4 of the textbook [RH04]. (See the syllabus for optional readings.)

Reading Problems

The problems in this section are intended to get you to read the textbook, ideally in advance of class meetings.

Read chapter 4, through section 4.1 of the textbook [RH04] and answer the following questions.

1. (5 points) [Concepts] [MapToLanguages]

The declarative concurrent model of chapter 4 adds threads to Oz. Threads are known to cause difficulties for reasoning about programs, because multiple threads can interleave their execution in many different orders.

These different execution orders allow observable nondeterminism in a language like Java or C#. The way that happens is that a program can use mutable state (such as cells) to record information that depends on the exact interleaving of each thread, and this can be used to give different outputs.

What property of Oz's declarative concurrent model makes this reasoning problem not possible in Oz?

2. [Concepts]

Consider the Oz statement in Figure 1.

```

local W X Y Z
in
  thread W = (Y-1)*10 end
  thread X = Y*10000 end
  thread Y = 7 end
  thread {Browse X+Y+W} end
end

```

Figure 1: Statement for problem Problem 2.

- (a) (3 points) What is shown in the browser after running this statement?
- (b) (2 points) Is it possible to see a different value if the threads execute in different orders?

Read chapter 4, through section 4.2 of the textbook [RH04] and answer the following questions.

3. [Concepts] [UseModels]

- (a) (3 points) Briefly explain how the concurrent map function example in section 4.2 (page 249) shows the use of dataflow behavior to give incremental output.
- (b) (3 points) If you change the code of the Map function in this example to no longer use the expression **thread** {F X} **end** but instead substitute the expression {F X}, does the code still have incremental behavior? Give a brief explanation.

Read chapter 4, through section 4.3 of the textbook [RH04] and answer the following questions.

4. (5 points) [Concepts] [MapToLanguages]

In what way is a Java (or C#) Iterator object like a stream in Oz?

5. [UseModels]

- (a) (3 points) What is a producer-consumer architecture?
- (b) (3 points) How can a producer-consumer architecture be implemented in Oz?

6. (5 points) [Concepts]

What is flow control? (Give a brief explanation.)

Read chapter 4, through section 4.4 of the textbook [RH04] and answer the following questions.

7. (5 points) [Concepts]

What problem can occur when programming with coroutines?

Read chapter 4, through section 4.5 of the textbook [RH04] and answer the following questions.

8. (9 points) [Concepts]

Suppose X is a dataflow variable that denotes a by-need suspension. For example, we might have executed the code

```
X = {ByNeed fun {$} {ExpensiveComputation} end}
```

Which of the following statements will cause Oz to determine the value of X? (Tell us all the correct answers, there may be more than one.)

- A. $Z = X + Y$
- B. `local Z in Z=X end`
- C. $X = 4020$
- D. `{proc {$ _} skip end X}`
- E. `if X then skip else skip end`

9. (5 points) [Concepts] [UseModels]

If you have a program that is organized using a producer-consumer architecture, which part should be made lazy to achieve flow control: the producer or the consumer? (Give a brief explanation.)

Read chapter 4, section 4.8 of the textbook [RH04] and answer the following questions. (You can skip section 4.6 and section 4.7.)

10. (5 points) [EvaluateModels]

Give an example of the kind of component or system that cannot be programmed easily and efficiently using the declarative concurrent model. Briefly explain why your example cannot be programmed in the declarative concurrent model.

Read chapter 4, section 4.9.2 of the textbook [RH04] and answer the following questions.

11. (3 points) [EvaluateModels] [MapToLanguages] Suppose you are designing a new Java-like programming language. Would it be a good idea to make all functions in your new language lazy by default?

Regular Problems

We expect you'll do the problems in this section after reading various parts of the chapter.

Some of the following problems are from the textbook [RH04, section 4.11].

Thread and Dataflow Behavior Semantics

The following problems explore the semantics of the declarative concurrent model.

12. (5 points) [Concepts]

Consider the code in problem 4 of section 4.11 of the textbook [RH04] (Order-determining concurrency). How does the Oz runtime system determine the order of execution of the assignment statements in this example? Briefly explain.

13. (0 points) [Concepts] [UseModels] (suggested practice)

Do problem 5 in section 4.11 of the textbook [RH04] (The Wait Operation).

14. (0 points) [Concepts] (suggested practice)

Do problem 8 in section 4.11 of the textbook [RH04] (Dataflow behavior in a concurrent setting).

Streams and Lazy Functional Programming

In the following the type $\langle \text{IStream } T \rangle$ means infinite lists of type T . Note that `nil` never occurs in a $\langle \text{IStream } T \rangle$, which has the following grammar:

$$\langle \text{IStream } T \rangle ::= T \text{ '}' \langle \text{IStream } T \rangle$$

15. (10 points) [UseModels]

Using the declarative concurrent model, write an incremental lazy function,

```
LSelect : <fun lazy {$ <IStream T> <fun {$ T}: Bool>}: <IStream T>>
```

that takes an infinite stream (i.e., an `<IStream>`), `IStream`, with elements of some type `T`, and a predicate `Pred`, that takes an element of type `T` and returns a Boolean. A call such as `{LSelect IStream Pred}` lazily returns an infinite stream that contains each element `E` of `IStream` for which `{Pred E}` is true, and no other elements. (The ordering of elements in the answer preserves the ordering seen in the input stream `IStream`.)

Be sure that your `LSelect` function does not loop forever when its input is infinite!

There are examples in Figure 2.

```
\insert 'LSelect.oz'
\insert 'TestingNoStop.oz'
declare % the next 3 functions are simply for use in testing
fun lazy {From N} N|{From N+1} end
fun {IsOdd N} (N mod 2) == 1 end
fun {IsEven N} (N mod 2) == 0 end
{Test {List.take {LSelect {From 1} IsOdd} 6} '==' [1 3 5 7 9 11]}
{Test {List.take {LSelect {From 1} IsEven} 9} '==' [2 4 6 8 10 12 14 16 18]}
{Test {List.take {LSelect {From 1} IsEven} 9} '==' [2 4 6 8 10 12 14 16 18]}
{Test {List.take {LSelect {From 100} IsEven} 5} '==' [100 102 104 106 108]}
{Test {List.take {LSelect {From 100} fun {$ E} E < 107 otherwise E == 999 end} 8}
      '==' [100 101 102 103 104 105 106 999]}
```

Figure 2: Tests for problem Problem 15.

16. (5 points) [Concepts] [UseModels] Is the function `LSelect` defined by your answer to Problem 15 incremental? (Say “yes, it is incremental” or “no, it’s not incremental.”) Briefly explain.

17. (10 points) [UseModels] Write a lazy function

```
RepeatingListOf : <fun lazy {$ <List T>}: <IStream T> >
```

that, for some type `T` takes a non-empty finite list `Elements` of elements of type `T`, and lazily returns an infinite stream (i.e., an `<IStream T>`) whose elements repeat the individual elements of `Elements` endlessly. See Figure 3 for examples.

```
\insert 'RepeatingListOf.oz'
\insert 'TestingNoStop.oz'
{Test {Nth {RepeatingListOf [1]} 1} '==' 1}
{Test {Nth {RepeatingListOf [7]} 500} '==' 7}
{Test {Nth {RepeatingListOf [2]} 999999} '==' 2}
{Test {List.take {RepeatingListOf [2 3 4]} 7} '==' [2 3 4 2 3 4 2]}
{Test {List.take {RepeatingListOf [a b c d e]} 11} '==' [a b c d e a b c d e a]}
{Test {List.take {RepeatingListOf "homework!"} 11} '==' "homework!ho"}
```

Figure 3: Tests for Problem 17.

18. (15 points) [UseModels]

Write a lazy function

```
BlockIStream: <fun lazy {$ <IStream <Char>> <Int>}: <IStream <List <Char> > >
```

that takes an infinite stream of characters, `IStream`, and an integer, `BlockSize`, and which lazily returns an infinite stream of lists of characters, where each list in the result has exactly `BlockSize` elements, consisting of the first `BlockSize` elements of `IStream`, followed by a list containing the next `BlockSize` elements of `IStream`, and so on. That is, a call to `BlockIStream` chunks the characters in `IStream` into strings of length `BlockSize`. (Hint: in your solution, you may want to use `List.take` and `List.drop`. Note that `{List.drop L N}` returns the list `L` without the first `N` elements.)

See Figure 4 for our tests.

```
% $Id: BlockIStreamTest.oz,v 1.3 2010/04/09 00:37:41 leavens Exp leavens $
\insert 'BlockIStream.oz'
\insert 'RepeatingListOf.oz' % from problem above, used here to make the tests
\insert 'TestingNoStop.oz'
declare
{StartTesting 'BlockIStream'}
fun lazy {From Char} Char|{From Char+1} end % for testing only
{TestLOS {List.take {BlockIStream {From &a} 3} % &a is the character a
  8}
  '==' ["abc" "def" "ghi" "jkl" "mno" "pqr" "stu" "vwx"]}
{TestLOS {List.take {BlockIStream {RepeatingListOf "Now is the time for..." } 1}
  12}
  '==' ["N" "o" "w" " " "i" "s" " " "t" "h" "e" " " "t"]}
{TestLOS {List.take {BlockIStream {RepeatingListOf "Now is the time for..." } 3}
  4}
  '==' ["Now" " is" " th" "e t"]}
{TestLOS {List.take {BlockIStream {RepeatingListOf "Now is the time for..." } 3}
  9}
  '==' ["Now" " is" " th" "e t" "ime" " fo" "r.." ".No" "w i"]}
{TestLOS {List.take {BlockIStream {RepeatingListOf "Now is the time for..." } 6}
  7}
  '==' ["Now is" " the t" "ime fo" "r...No" "w is t" "he tim" "e for."]}
{StartTesting done}
```

Figure 4: Tests for Problem 18 on the previous page.

19. (15 points) [UseModels]

Write a lazy function

```
EncryptIStream: <fun lazy {$ <IStream <Char>> <Int> <fun {$ <List <Char> >}: <List <Char> >}
               : <IStream <List <Char> > >
```

that takes 3 arguments: an infinite stream of characters, `IStream`, an integer, `BlockSize`, and a function, `Encrypt`. The `EncryptIStream` function lazily returns an infinite stream of lists of characters, where each list in the result is the result of applying `Encrypt` to a list containing `BlockSize` elements of `IStream`. The first list in the result is the result of applying `Encrypt` to the first `BlockSize` elements of `IStream`, and this is followed by the result of applying `Encrypt` to the next `BlockSize` elements of `IStream`, and so on. That is, a call to `EncryptIStream` first chunks the characters in `IStream` into strings of length `BlockSize`, and then applies `Encrypt` to each resulting list of strings. Figure 5 on the following page shows some examples, written using various (cryptographically poor) `Encrypt` functions.

```
\insert 'BlockIStream.oz' % So you can use BlockIStream in your answer.
```

20. (3 points) [Concepts] Does your function `EncryptIStream` in question 19 have to be lazy? Answer “yes” or “no” and give a brief explanation.

```

\insert 'EncryptIStream.oz'
\insert 'RepeatingListOf.oz' % from problem above, used here to make the tests
\insert 'TestingNoStop.oz'
declare
fun {NoEncryption Str} Str end
fun {ReverseNoEncryption Str} {Reverse {NoEncryption Str}} end
fun {CeaserCypher Str} {Map Str fun {$ C}{C+1 mod 512 end} end
fun {ReverseCeaserCypher Str} {Reverse {CeaserCypher Str}} end
{Test {List.take {EncryptIStream {RepeatingListOf "Now is the time for..."} 12 NoEncryption}
  3}
'==' ["Now is the t" "ime for...No" "w is the tim" ]}
{Test {List.take {EncryptIStream {RepeatingListOf "We're off to see the Wizard, the Wonderful Wizard of Oz"}
  3 ReverseNoEncryption}
  18}
'==' ["'eW" " er" "ffo" "ot " "es " "t e" " eh" "ziW" "dra" "t ," " eh"
  "noW" "red" "luf" "iW " "raz" "o d" "O f"]}
{Test {List.take {EncryptIStream {RepeatingListOf "Now is the time for..."} 5 CeaserCypher}
  7}
'==' ["OpX!j" "t!uif" "!ujnf" "!gps/" "//OpX" "!jt!u" "if!uj"]}
{Test {List.take {EncryptIStream {RepeatingListOf "Now is the time for..."} 5 ReverseCeaserCypher}
  7}
'==' ["j!xp0" "fiu!t" "fnju!" "/spg!" "xp0/" "u!tj!" "ju!fi"]}

```

Figure 5: Tests for Problem 19 on the previous page.

Laziness Problems

The following problems explore more about laziness and its utility.

21. (20 points) [Concepts] [UseModels]

Do problem 16 in section 4.11 of the textbook [RH04] (By-need execution).

To explain this problem and provide a test, call your procedure that solves this problem `RequestCalc`. Note that it should be a procedure, not a function. Then when run as in Figure 6 it should show in the Browser window first `Z`, then it should show `requestin`, then the `Z` should change to `0|_`, then you should see `request2in`, and then the first line should change to `0|1|_`. At some point you will also see `A` and then `requestAin` followed by `done`, then the `A` should change to `an_atom`.

```
% $Id: RequestCalcTest.oz,v 1.3 2010/04/11 14:50:41 leavens Exp leavens $
\insert 'TestingNoStop.oz'
\insert 'RequestCalc.oz'

% Testing
declare Z
fun lazy {SGen Y} {Delay 5000} Y|{SGen Y+1} end
Z = {SGen 0}
{Browse Z}

{Delay 2000} {RequestCalc Z} {Browse requestin}
{Delay 2000} {RequestCalc Z.2} {Browse request2in}

declare X
fun lazy {LThree} {Delay 5000} 3 end
X = {LThree}
{Browse X}

{Delay 2000} {RequestCalc X} {Browse requestXin}

fun lazy {LAtom} {Delay 5000} an_atom end
A = {LAtom}
{Browse A}
{Delay 2000} {RequestCalc A} {Browse requestAin}
{Browse done}
```

Figure 6: Tests for Problem 21.

Hints: think about what actions in Oz request (demand) calculation of a variable identifier's value. And think about what new features we have in this chapter that can be used to prevent a thread from waiting for something.

The following problems, inspired by a paper written by John Hughes, relate to modularization of numeric code using streams and lazy execution. In particular, we will explore numerical integration.

As an aid to writing code for this section, and for testing that code, we provide a library file containing predicates for approximate comparisons of floating point numbers and for testing with approximate comparisons. The floating point approximate comparison code is shown in Figure 7 on the following page. The testing code for floating point numbers is shown in Figure 8 on page 9.

22. (10 points) [UseModels] Write a lazy function

```
IStreamIterate: <fun lazy {$ <fun {$ T}: T> T}: <IStream T>>
```

```

% $Id: FloatPredicates.oz,v 1.3 2007/10/23 02:14:21 leavens Exp leavens $
% Some functions to do approximate equality of floating point numbers.
% AUTHOR: Gary T. Leavens

declare
%% Return true iff the difference between X and Y
%% is no larger than Epsilon
fun {Within Epsilon X Y} {Abs X-Y} =< Epsilon end

%% Partly curried version of Within
fun {WithinMaker Epsilon} fun {$ X Y} {Within Epsilon X Y} end end

%% Return true iff the corresponding lists are
%% equal relative to the given predicate
fun {CompareLists Pred Xs Ys}
  case Xs#Ys of
    nil#nil then true
    [] (X|Xr)#(Y|Yr) then {Pred X Y} andthen {CompareLists Pred Xr Yr}
    else false
  end
end

%% Return true iff the lists are equal
%% in the sense that the corresponding elements
%% are equal to within Epsilon
fun {WithinLists Epsilon Xs Ys}
  {CompareLists {WithinMaker Epsilon} Xs Ys}
end

%% Return true iff the ratio of X-Y to Y is within Epsilon
fun {Relative Epsilon X Y} {Abs X-Y} =< Epsilon*{Abs Y} end

%% Partly curried version of Relative
fun {RelativeMaker Epsilon} fun {$ X Y} {Relative Epsilon X Y} end end

%% Return true iff the lists are equal
%% in the sense that the corresponding elements
%% are relatively equal to within Epsilon
fun {RelativeLists Epsilon Xs Ys}
  {CompareLists {RelativeMaker Epsilon} Xs Ys}
end

%% A useful tolerance for testing
StandardTolerance = 1.0e~3

%% A convenience for testing, relative equality with a fixed Epsilon
ApproxEqual = {RelativeMaker StandardTolerance}

```

Figure 7: Comparisons for floating point numbers. This code is available in this homework's zip file and also in the course lib directory.


```

% $Id: FloatTesting.oz,v 1.5 2008/03/24 15:43:04 leavens Exp leavens $
% Testing for floating point numbers.
% AUTHOR: Gary T. Leavens

\insert 'FloatPredicates.oz'
\insert 'TestingNoStop.oz'

declare
%% TestMaker returns a procedure P such that {P Actual '=' Expected}
%% is true if {FloatCompare Epsilon Actual Expected} (for Floats)
%% or if {FloatListCompare Epsilon Actual Expected} (for lists of Floats)
%% If so, print a message, otherwise throw an exception.
fun {TestMaker FloatCompare FloatListCompare Epsilon}
  fun {Compare Actual Expected}
    if {IsFloat Actual} andthen {IsFloat Expected}
    then {FloatCompare Epsilon Actual Expected}
    elseif {IsList Actual} andthen {IsList Expected}
    then {FloatListCompare Epsilon Actual Expected}
    else false
    end
  end
in
  proc {$ Actual Connective Expected}
    if {Compare Actual Expected}
    then {System.showInfo
      {Value.toVirtualString Actual 5 20}
      # ' ' # Connective # ' '
      # {Value.toVirtualString Expected 5 20}}
    else {System.showInfo
      'TEST FAILURE: '
      # {Value.toVirtualString Actual 5 20}
      # ' ' # Connective # ' '
      # {Value.toVirtualString Expected 5 20}
      }
    end
  end
end

WithinTest = {TestMaker Within WithinLists StandardTolerance}
RelativeTest = {TestMaker Relative RelativeLists StandardTolerance}

```

Figure 8: Testing code for floating point. This puts output on standard output (the *Oz Emulator* window). The file FloatPredicates is shown in Figure 7 on the previous page. This file is available in this homework's zip file and also in the course lib directory.

such that `{IStreamIterate F X}` takes a function F and a value X and returns the lazy infinite stream

$$X \mid \{FX\} \mid \{F\{FX\}\} \mid \{F\{F\{FX\}\}\} \mid \dots,$$

that is, the stream whose i^{th} item, counting from 1, is F^{i-1} applied to X .

The examples in Figure 9 on the following page are written using the `WithinTest` procedure from Figure 8 on the previous page. Notice also that, since the function `Next` in the testing file is curried, we don't pass `Next` itself to `IStreamIterate`, but instead pass the value of applying `Next` to some number.

```

% $Id: IStreamIterateTest.oz,v 1.2 2010/03/24 21:26:54 leavens Exp leavens $
\insert 'IStreamIterate.oz'
\insert 'FloatTesting.oz'
declare
% Next: <fun {$ <Float>}: <fun {$ <Float>}: Float>>
fun {Next N}
  fun {$ X}
    (X + N / X) / 2.0
  end
end
{StartTesting 'IStreamIterate...'}
{WithinTest {List.take {IStreamIterate {Next 1.0} 1.0} 7}
  '~::~' [1.0 1.0 1.0 1.0 1.0 1.0 1.0]}
{WithinTest {List.take {IStreamIterate {Next 9.0} 1.0} 7}
  '~::~' [1.0 5.0 3.4 3.0235 3.0001 3.0 3.0]}
{WithinTest {List.take {IStreamIterate {Next 200.0} 1.0} 7}
  '~::~' [1.0 100.5 51.245 27.574 17.414 14.449 14.145]}
{RelativeTest {List.take {IStreamIterate {Next 0.144} 7.0} 9}
  '~::~' [7.0 3.5103 1.7757 0.92838 0.54174 0.40378 0.3802 0.37947 0.37947]}
{RelativeTest {List.take {IStreamIterate fun {$ X} X*X end 2.0} 9}
  '~::~' [2.0 4.0 16.0 256.0 65536.0 4.295e009 1.8447e019 3.4028e038
    1.1579e077]}
{RelativeTest {List.take {IStreamIterate fun {$ X} X/3.0 end 10.0} 8}
  '~::~' [10.0 3.3333 1.1111 0.37037 0.12346 0.041152 0.013717 0.0045725]}
{StartTesting 'done'}

```

Figure 9: Tests for Problem 22 on page 7.

23. (10 points) [UseModels]

Write a function

ConvergesTo: `<fun {$ <IStream T> <fun {$ T T}: Bool>}: T>`

such that `{ConvergesTo Xs Pred}` looks down the stream Xs to find the first two consecutive elements of Xs that satisfy Pred, and it returns the second of these consecutive elements. (It will never return if there is no such pair of consecutive elements.) Figure 10 gives some examples.

```
% $Id: ConvergesToTest.oz,v 1.1 2009/02/28 20:52:54 leavens Exp $
\insert 'ConvergesTo.oz'
\insert 'FloatTesting.oz'

fun lazy {Repeat X} X|{Repeat X} end

{StartTesting 'ConvergesTo'}
{WithinTest {ConvergesTo
  {Append [1.0 3.5 4.5] {Repeat 7.0}}
  {WithinMaker 1.01}
}
'~~' 4.5}
{WithinTest {ConvergesTo
  {Append [1.0 32.5 17.2346 10.474 8.29219 8.00515] {Repeat 8.0}}
  {WithinMaker 0.5}
}
'~~' 8.00515}
{StartTesting 'done'}
```

Figure 10: Tests for Problem 23.

24. (15 points) [UseModels]

You may recall that the derivative of a function F at a point X can be approximated by the following function.

```
% $Id: EasyDiff.oz,v 1.3 2007/10/31 00:44:27 leavens Exp leavens $
declare
fun {EasyDiff F X Delta} ({F (X+Delta)} - {F X}) / Delta end
```

Good approximations are given by small values of Delta, but if Delta is too small, then rounding errors may swamp the result. One way to choose Delta is to compute a sequence of approximations, starting with a reasonably large one. (In this way, if `{WithinMaker Epsilon}` is used to select the first approximation that is accurate enough, this can reduce the risk of a rounding error affecting the result, as we will show in the next problem.)

In this problem your task is to write a function

DiffApproxims: `<fun {$ Float <fun {$ Float}: Float> Float}: <IStream Float>>`

such that `{DiffApproxims Delta0 F X}` returns an infinite list of approximations to the derivative of F at X, where at each step, the current Delta is halved. Examples are given in Figure 11 on the next page.

Hint: do you need to make something lazy to make this work?

25. (15 points) [UseModels]

Using the pieces given above, in particular `ConvergesTo` and `DiffApproxims`, write a function

Differentiate: `<fun {$ Float Float <fun {$ Float}: Float> Float}: Float>`

```

% $Id: DiffApproximsTest.oz,v 1.1 2009/02/28 20:52:54 leavens Exp $
\insert 'DiffApproxims.oz'
\insert 'FloatTesting.oz'
{StartTesting 'DiffApproxims'}
{WithinTest {List.take {DiffApproxims 500.0 fun {$ X} X*X end 20.0} 9}
  '~::~' [540.0 290.0 165.0 102.5 71.25 55.625 47.8125 43.9062 41.9531]}
{WithinTest {List.take {DiffApproxims 100.0 fun {$ X} X*X*X end 10.0} 8}
  '~::~' [13300.0 4300.0 1675.0 831.25 526.562 403.516 349.316 324.048]}
{StartTesting 'done'}

```

Figure 11: Tests for Problem 24 on the preceding page.

such that $\{\text{Differentiate } \epsilon \Delta_0 F N\}$ returns an approximation that is accurate to within ϵ to the derivative of F at N . Use the previous problem (Problem 24 on the previous page) with Δ_0 as the initial value for Δ . Examples are given in Figure 12.

```

% $Id: DifferentiateTest.oz,v 1.2 2009/02/28 20:52:54 leavens Exp $
\insert 'FloatTesting.oz'
\insert 'Differentiate.oz'
{StartTesting 'Differentiate'}
Millionth = 0.0000001
WithinMillionth = {TestMaker
  Within
  WithinLists
  Millionth}
{WithinMillionth {Differentiate Millionth 500.0 fun {$ X} X*X end 20.0}
  '~::~' 40.0}
{WithinMillionth {Differentiate Millionth 100.0 fun {$ X} X*X*X end 10.0}
  '~::~' 300.0}
{StartTesting 'done'}

```

Figure 12: Tests for Problem 25 on the previous page.

Problems and Programming Models

The following problems ask you to compare different programming models and the problems they are good at solving.

26. (20 points) [EvaluateModels]

Make a table listing all the different programming techniques, the characteristics of problems that are best solved with these techniques (i.e., when to use the techniques), and the name of at least one example of that technique.

programming technique	problem characteristics	example(s)
recursion (over grammars)		
higher-order functions		
stream programming		
lazy functions		

Extra Credit Problems

Extra credit problems are entirely optional. See the course grading policy for details.

27. (20 points; extra credit) [Concepts]

Do problem 7 in section 4.11 of the textbook [RH04] (Programmed triggers using higher-order programming). Note that the example mentioned in the problem is the example with `DGenerate` and `DSum` that is the first example shown in section 4.3.3.1.

28. (15 points; extra credit) [Concepts]

Do problem 18 in section 4.11 of the textbook [RH04] (Concurrency and Exceptions). (Hint: the “abstract machine semantics” is the operational semantics of Oz, which is described in the textbook’s sections 2.4, 2.7.2, 4.1, and 4.9.1. You don’t have to do anything formal or especially mathematical with the operational semantics.)

29. (40 points; extra credit) [Concepts]

The instructions for this homework state that you are not to use Oz’s built-in function `IsDet` in writing your code, since that is not part of the declarative model. Show why the use of `IsDet` takes you outside the declarative model by using it to simulate a Boolean-valued cell. (Of course, you are to do this only using the declarative model and `IsDet`. In particular you must not use cells in your solution!) Let’s call the ADT that you are programming `CellWithIsDet`. Then your code should have the following interface.

```
NewCellWithIsDet: <fun {$ Bool}: CellWithIsDet>
CWIDAccess: <fun {$ CellWithIsDet}: Bool>
CWIDAssign: <proc {$ CellWithIsDet Bool}>
```

Testing should show that you can change the value of a `CellWithIsDet` object from true to false and back again, several times, as shown in Figure 13 on the following page.

(Warning: this problem is decidedly non-trivial! That’s why it’s extra credit!)

Points

This homework’s total points: 212. Total extra credit points: 75.

References

- [CW98] Mary Campione and Kathy Walrath. *The Java Tutorial Second Edition: Object-Oriented Programming for the Internet*. The Java Series. Addison-Wesley, Reading, MA, second edition, 1998.
- [DKS06] Denys Duchier, Leif Kornstaedt, and Christian Schulte. *The Oz Base Environment*. moztart-oz.org, June 2006. Version 1.3.2.
- [RH04] Peter Van Roy and Seif Haridi. *Concepts, Techniques, and Models of Computer Programming*. The MIT Press, Cambridge, Mass., 2004.

```

% $Id: CellWithIsDetTest.oz,v 1.1 2009/02/28 20:52:54 leavens Exp $
\insert 'TestingNoStop.oz'
\insert 'CellWithIsDet.oz'

{StartTesting 'CellWithIsDet'}
declare
MyCWID = {NewCellWithIsDet true}
Alias = MyCWID
MyCWID2 = {NewCellWithIsDet false}
{Test {CWIDAccess MyCWID} '==' true}
{Test {CWIDAccess Alias} '==' true}
{Test {CWIDAccess MyCWID2} '==' false}
{CWIDAssign MyCWID false}
{Test {CWIDAccess MyCWID} '==' false}
{Test {CWIDAccess Alias} '==' false}
{Test {CWIDAccess MyCWID2} '==' false}
{CWIDAssign MyCWID2 true}
{Test {CWIDAccess MyCWID} '==' false}
{Test {CWIDAccess Alias} '==' false}
{Test {CWIDAccess MyCWID2} '==' true}
{CWIDAssign MyCWID2 true}
{Test {CWIDAccess MyCWID} '==' false}
{Test {CWIDAccess Alias} '==' false}
{Test {CWIDAccess MyCWID2} '==' true}
{CWIDAssign MyCWID2 false}
{Test {CWIDAccess MyCWID2} '==' false}
{CWIDAssign MyCWID2 false}
{Test {CWIDAccess MyCWID2} '==' false}
{CWIDAssign MyCWID2 true}
{Test {CWIDAccess MyCWID2} '==' true}
{StartTesting 'done'}

```

Figure 13: Testing for Problem 29 on the previous page.